```
> with(OreModules):
> with(OreMorphisms):
> with(Stafford):
> with(linalg):
```

Let us consider the third Weyl algebra $A = A_3(\mathbb{Q})$, where $\mathbb{Q}$ is the field of rational numbers,

```
> A := DefineOreAlgebra(diff=[d[3],x[3]], diff=[d[1],x[1]],
> diff=[d[2],x[2]], polynom=[x[3],x[1],x[2]]):
```

and the left A-module L finitely presented by the matrix P of PD operators defining the curl operator in 3-space, namely:

```
> P := evalm([[0,-d[3],d[2]],[d[3],0,-d[1]],[-d[2],d[1],0]]);
```

$$P := \begin{bmatrix} 0 & -d_3 & d_2 \\ d_3 & 0 & -d_1 \\ -d_2 & d_1 & 0 \end{bmatrix}$$

Let us compute the rank of L:

```
> OreRank(P,A);
```

$$1$$

Thus, we get $rank_A(\mathrm{L}) = 1$. We can check that L is not a free left A-module of rank 1 since $ext_A{}^3(\mathrm{N}, A)$ is isomorphic to A / ($A\, d_1$ + $A\, d_2$ + $A\, d_3$), which is non-zero, where N is the Auslander transpose of L:

```
> Exti(Involution(P,A),A,2);
```

$$\left[\begin{bmatrix} d_2 \\ d_1 \\ d_3 \end{bmatrix}, \begin{bmatrix} 1 \end{bmatrix}, \mathrm{SURJ}(1)\right]$$

Equivalently, we can check that no generalized inverse of P exists:

```
> GeneralizedInverse(P,A);
```

$$[]$$

Hence, L can be generated by two elements. Using Stafford's reduction, let us try to compute a presentation of L with two generators:

```
> S := map(collect,StaffordReduction(P,A),[d[1],d[2],d[3]]):
```

The output of the command *StaffordReduction* is a list with two entries.

```
> nops(S);
```

$$2$$

The first entry S[1] of S

```
> S[1];
```

$$\begin{bmatrix} 0 & 0 \\ -{d_3}^2 & -d_1\, d_3 + (2 + d_3\, (x_3 + 1))\, d_2 \\ d_1\, d_3 + (1 + (-x_3 - 1)\, d_3)\, d_2 & {d_1}^2 + (-2\, x_3 - 2)\, d_2\, d_1 + (2\, x_3 + {x_3}^2 + 1)\, {d_2}^2 \end{bmatrix}$$

is a matrix presenting a left A-module L_2 isomorphic to L. The second entry S[2] of S defines this left A-isomorphism

```
> S[2];
```

$$\begin{bmatrix} -1 & x_3+1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

i.e., the left A-homomorphism γ from L_2 to L induced by S[2] is a left A-isomorphism:

```
> TestIso(S[1],P,S[2],A);
```

$$\mathit{true}$$

A minimal set of generators of L is defined by the images under γ of the residue classes represented by (1 0) and (0 1), i.e.,

$$z_1 = -\, y_1 + (\, x_3 + 1)\; y_2, \quad z_2 = y_3,$$

where { y_1, y_2, y_3 } is a set of generators of L. Moreover, computing γ^{-1},

```
> U := InverseMorphism(S[1],P,S[2],A);
```

$$U := \left[\begin{bmatrix} d_3\,x_3 + d_3 - 1 & -2\,d_2\,x_3 - d_2\,{x_3}^2 - d_2 + d_1\,x_3 + d_1 \\ d_3 & -d_2\,x_3 - d_2 + d_1 \\ 0 & 1 \end{bmatrix}, \begin{bmatrix} 0 & 1 & 0 \\ 0 & -x_3 - 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}\right]$$

we obtain the left A-isomorphism γ^{-1} induced by U[1]. Thus, in terms of generators, we get y = U[1] z, where y = (y_1, y_2, y_3)ˆT, z = (z_1, z_2)ˆT.

Let us try to reduce the number of relations of the above presentation of L_2.

```
> T := map(collect,StaffordReduction(P,A,"reduce_relations"=true),
> [d[1],d[2],d[3]]):
```

We obtain that L_2 is finitely presented by T[1], where T[1] is defined by

```
> T[1];
```

$$\begin{bmatrix} -d_1\,d_3 + (-1 + d_3\,(x_3+1))\,d_2 & -{d_1}^2 + (2\,x_3+2)\,d_2\,d_1 + (-2\,x_3 - {x_3}^2 - 1)\,{d_2}^2 \\ -{d_3}^2 & -d_1\,d_3 + (2 + d_3\,(x_3+1))\,d_2 \end{bmatrix}$$

and is isomorphic to L, i.e., the left A-isomorphism γ from L_2 to L is induced by T[2], where T[2] is defined by

```
> T[2];
```

$$\begin{bmatrix} -1 & x_3+1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

i.e., T[2] = S[2]. To conclude, the linear PD system P y = 0 defined by the curl operator in 3-space is equivalent to the linear PD system T[1] (z_1, z_2)ˆT = 0 in two unknowns z_1, z_2 and two equations.